

```
log using "F:\workshop\long format\event_05_21_18.log", replace
```

```
use "F:\workshop\long format\command_long.dta", clear
```

```
des
```

```
list, sepby(id)
```

```
*****  
* Sort data  
*****
```

```
sort id year month
```

```
*****  
* Create an indicator variable for each timepoint within the person  
*****  
by id: gen time = _n  
label variable time "time points"
```

```
*****  
*Calculate the total number of records per person  
*****  
by id: gen t_time = _N  
label variable t_time "total time points"
```

```
list, sepby(id)
```

```
*****  
* Check whether each record is a unique one  
*****
```

```
duplicates report id year month
```

```
*****  
* Reshape data: from long format to wide format  
*****
```

```
reshape wide year month marst sex age t_time, i(id) j(time)
```

```
order id year* month* marst* sex* age* t_time*
```

```
list year* if id ==1  
list month* if id ==1  
list marst* if id ==1  
list sex* if id ==1  
list age* if id ==1  
list t_time* if id ==1
```

```
list year* if id ==2  
list month* if id ==2  
list marst* if id ==2  
list sex* if id ==2  
list age* if id ==2  
list t_time* if id ==2
```

```
*****  
* Reshape data: from long format to wide format  
*****
```

```
reshape long year month marst sex age t_time, i(id) j(time)
```

```
list, sepby(id)  
drop if t_time ==.  
list, sepby(id)
```

```

*****
* sort: Arrange data in a special order
*****

sort id time
list, sepby(id)

*****
* Handling the missing value
*****

list id time year month marst, sepby(id)

gen marst_r = marst
label variable marst_r "recoded marital status"

replace marst_r = marst[_n-1] if time ~=1 & marst_r ==. & marst_r[_n-1] ~=.

list id time year month marst marst_r, sepby(id)

*****
* Code the transition into marriage
*****

gen c_mar =0
label variable c_mar "change into marriage"

by id: replace c_mar = 1 if marst_r[_n-1] >=4 & (marst_r ==1 | marst_r ==2| marst_r ==3 )

list id time year month marst_r c_mar, sepby(id)

*****
*Create an indicator for the times that respondents entered marriage
*****

by id: gen i_c_mar = sum(c_mar)
label variable i_c_mar "indicator for the times of entering marriage"

*****
* Create an indicator for the total number of times of entering marriage"
*****

by id: egen s_c_mar = sum(c_mar)
label variable s_c_mar "indicator for total number of times of entering marriage"

list id time year month marst_r c_mar i_c_mar s_c_mar, sepby(id)

*****
*Extract the time when the first marriage took place
*****

by id: gen time_mar1 = time if c_mar ==1 & i_c_mar ==1

label variable time_mar1 "time for the first marriage"

list id time year month marst_r c_mar i_c_mar s_c_mar time_mar1, sepby(id)

*****
* Expand the time of the first transition into marriage for all records of the individual

```

```
by id: egen m_time_mar1 = max(time_mar1)
```

```
label variable m_time_mar1 "time for the first marriage"
```

```
list id time year month marst_r c_mar i_c_mar s_c_mar time_mar1 m_time_mar1, sepby(id)
```

```
* Remove records that occurred after the first transition into marriage
```

```
list id time year month marst_r c_mar i_c_mar s_c_mar time_mar1 m_time_mar1, sepby(id)
```

```
by id: drop if time > m_time_mar1
```

```
list id time year month marst_r c_mar i_c_mar s_c_mar time_mar1 m_time_mar1, sepby(id)
```

```
* Save the event history data
```

```
list id time year month marst_r sex age, sepby(id)
```

```
save "F:\workshop\long format\event.dta", replace
```

```
log close
```